Integrationstechniken

Aufgaben: 3

Obschon MAPLE gute Routinen hat, um Integrale bzw. Stammfunktionen zu bestimmen, muß man doch die beiden wesentlichen Techniken kennen, wie man dies von Hand machen kann, da diese auch in theoretischen Betrachtungen eine Rolle spielen kann. Übrigens ist die Frage, ob eine sogenannte elementare Funktion auch ein elementares Integral hat, ein algebraisches Problem, das algorithmisch entschieden werden kann (Risch-Algorithmus, vgl. Bronstein: Symbolic Integration I, Springer). In MAPLE sind die wichtigsten Teile des Algorithmus implementiert.

```
> restart;
```

Partielle Integration

MATH: Eine der wichtigsten Regeln für die Differentiation ist die Produktregel:

$$(uv)' = u'v + uv'$$

für $u, v \in \mathrm{C}^1([a, b])$.

Mit Hilfe des Fundamentalsatzes ergibt sich hieraus sofort die **partielle Integration** (**DENKANSTOSS**: Führe dies formal aus.):

$$\int_a^x u'(t)v(t)dt = u(x)v(x) - u(a)v(a) - \int_a^x u(t)v'(t)dt$$

```
> Int(diff(u(t),t)*v(t),t=a..x)= u(x)*v(x) - u(a)*v(a)- Int(u(t)*diff(v(t),t),t=a..x);
```

$$\int_a^x \left(\frac{\mathrm{d}}{\mathrm{d}t} u(t)\right) v(t)\, \mathrm{d}t = u(x)\, v(x) - u(a)\, v(a) - \left(\int_a^x u(t) \left(\frac{\mathrm{d}}{\mathrm{d}t} v(t)\right) \mathrm{d}t\right) \quad \textbf{(1.1.1)}$$

für alle $x \in [a, b]$. Wenn man nur von Stammfunktionen spricht, schreibt man meistens nur

$$\int^x u'(t)v(t)dt = u(x)v(x) - \int^x u(t)v'(t)dt$$

Eine typische Anwendung der partiellen Integration ist der Abbau von Polynomfaktoren bei der Integration von Produkten von Polynomfunktionen mit anderen leicht integrierbaren Funktionen:

Beispiel:

```
> int(exp(x)*x,x);
```

$$(x-1)\, \mathrm{e}^x \quad \textbf{(1.1.2)}$$

MATH: Wenn es mit dem Abbau von Faktoren nicht funktioniert, so kann man hoffen, daß nach ein- oder mehrmaliger Anwendung der partiellen Integration ein Vergleich möglich ist.

ÜBUNG [01]:

1) Integriere $x \to x^2 \cdot \sin(x)$, ohne den **int**-Befehl auf eine Produktfunktion anzuwenden.
2) Integriere die Funktion $\sin(x) \cdot \exp(x)$ (nach dem obigen Hinweis), ohne den **int**-Befehl auf eine Produktfunktion anzuwenden. Interpretiere den Faktor 1/2.
(Hinweis: Da man mehrfach partiell integrieren muss, lohnt es sich schon die partielle Integrationsformel als Programm zu schreiben.)

Integration durch Substitution

MATH: Eine weitere sehr wichtige Differentiationsregel ist die Kettenregel:

$\frac{du(v(x))}{dx} = u'(v(x)) \cdot v'(x)$ oder $(u \circ v)' = (u' \circ v) \cdot v'$

```
> diff(u(v(x)),x);
```

$$\mathrm{D}(u)(v(x)) \left(\frac{\mathrm{d}}{\mathrm{d}x} v(x)\right) \quad (1.2.1)$$

Dieser steht gemäß dem Fundamentalsatz die Substitutionsregel beim Integrieren gegenüber:

$$\int^{x} u(v(t))\, v'(t)\, dt = \int^{v(x)} u(y)\, dy$$

oder

$$\int_a^x u'(v(t))\, v'(t)\, dt = u(v(x)) - u(v(a))$$

(**DENKANSTOSS**: Führe dies formal aus.)

```
> Int(diff(u(v(t)),t),t=a..x)=u(v(x))-u(v(a));
```

$$\int_a^x \mathrm{D}(u)(v(t)) \left(\frac{\mathrm{d}}{\mathrm{d}t} v(t)\right) \mathrm{d}t = u(v(x)) - u(v(a)) \quad (1.2.2)$$

Einfache Beispiele:

```
> int(sin(a*x),x);
```

$$-\frac{\cos(a\,x)}{a} \quad (1.2.3)$$

```
> int(sin(x^2)*x,x);
```

$$-\frac{1}{2}\cos(x^2) \quad (1.2.4)$$

```
> int(sin(cos(x))*sin(x),x);
```

(1.2.5)

$$\cos(\cos(x)) \quad \textbf{(1.2.5)}$$

ÜBUNG [02]:

Integriere $x \rightarrow \frac{x \cdot \cos(x^2)}{\sin(x^2)^{10}}$ unter Benutzung der Substitutionsregel. (Vorsicht bei Nullstellen im Nenner!)

Integration der Umkehrfunktion

Wir brauchen ein Hilfsprogramm:

```
> #Funktioniert nur mit neueren Maple-Version
  FillBetweenCurves:=proc(f1, f2, {[color, colour] := red,
  transparency := 0.0})
> local p, c, n, pts1, pts2, polys, i, x, ytop, ybottom,
  yvals;
>  p:=plot([f1, f2], _rest, _options['color'],numpoints=51,
  adaptive=false);
>  c:=select(type, p, 'specfunc'('anything', 'CURVES'));
>  pts1:=op([1,1], c);pts2:=op([2,1], c);
   n:=LinearAlgebra[RowDimension](pts1);
>  polys:=NULL;x:=Vector(n, datatype=float);ytop:=Vector(n,
  datatype=float);ybottom:=Vector(n, datatype=float);
>  for i to n do
>    x[i]:=pts1[i,1];
>    yvals:=pts1[i,2],pts2[i,2];
>    ytop[i]:=max(yvals);
>    ybottom[i]:=min(yvals);
>  end do;
>  polys := [seq([x[i], ytop[i]], i=1..n), seq([x[i], ybottom
  [i]], i=n..1, -1),[x[1], ytop[1]]];
>  plots[display](plottools[polygon](polys, _options
  ['color'],_options['transparency'],filled=true), p);
> end proc:
```

Wir haben bereits die Ableitung einer Umkehrfunktion untersucht. Nun soll es um Integration der Umkehrfunktion gehen.

MATH: (Integration der Umkehrfunktion) Sei f: $[a, b] \rightarrow \mathbb{R}$ stetig und bijektiv und g die Umkehrfunktion von f. Dann gilt:

$$\int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y = b \cdot f(b) - a \cdot f(a) - \int_a^b f(x)\,\mathrm{d}x$$

Wir deuten die Idee der Formel an einem Beispiel an. Sei dazu $f\colon [a, b] \to \mathbb{R}_{>0} : x \mapsto x \cdot e^x$ mit $a = \frac{1}{2}$ und $b = 1$. Man sieht leicht ein, dass f injektiv (und damit bijektiv auf sein Bild) ist, indem man die Ableitung von f bildet.

```
> f:=x->exp(x)*x;
  a:=1/2:
  b:=1:
```

$$f := x \to e^x x \tag{1.3.1}$$

Wir kennen bereits die rote Fläche unter dem Graphen von f in dem Intervall $[a, b]$, da wir durch partielle Integration die Stammfunktion von f gefunden haben:

```
> plots[display](FillBetweenCurves(0,f(x),x=a..b,color=red,
  transparency=0.25,scaling=constrained,color=red,thickness=2,
  filled=true),plot(f,0..b,scaling=constrained,color=black,
  thickness=2));
```



Wir können aber noch nicht die Stammfunktion der Umkehrfunktion g von f berechnen :

```
> g:=unapply(solve(f(x)=y,x),y);
```

$$g := y \to \mathrm{LambertW}(y) \qquad \textbf{(1.3.2)}$$

Wir nutzen jetzt einen Trick, um den Inhalt

$$\int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y$$

der blauen Flächen im folgenden Plot zu bestimmen.

```
> plots[display](FillBetweenCurves(max(f(a),f(x)),f(b),x=0..b,
  color=blue,transparency=0.5,scaling=constrained),
  FillBetweenCurves(0,f(x),x=a..b,color=red,transparency=0.5,
  scaling=constrained),plot(f,0..b,color=black,thickness=2));
```



Die Vereinigung der roten und der blauen Fläche ist gleich

$$\int_a^b f(x)\,\mathrm{d}x + \int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y = b \cdot f(b) - a \cdot f(a),$$

wobei $a \cdot f(a)$ der Flächeninhalt des weißen Rechteckes links unten ist und $b \cdot f(b)$ der Flächeninhalt des gesamten Plots.
Hier können wir das erste Integral und die komplette rechte Seite ausrechnen.

DENKANSTOSS: Die Formel läßt sich auch direkt aus der Produktregel und Substitutionsregel herleiten. Idee:

$$\int x\,\mathrm{d}y = x \cdot y - \int y\,\mathrm{d}x$$

ÜBUNG [03]:

Bestimme $\int_0^1 \arcsin(x)\,\mathrm{d}x$ ohne den **int**-Befehl zu verwenden.

Integrationstechniken: Partialbruchzerlegung

Aufgaben: 3

```
> restart;
```

Inhalt

Wir hatten bereits die partielle Integration und die Substitutionsregel behandelt. Hier soll die Rede von der Partialbruchzerlegung sein, die rationale Funktionen so vereinfacht, daß man sie möglicherweise leichter integrieren kann. Wir arbeiten über dem Körper $\mathbb{R}$ der reellen Zahlen.

MATH: Eine rationale Funktion ist der Quotient zweier Polynomfunktionen. Möglicherweise hat selbst im durchgekürzten Fall der Nenner noch Nullstellen, so daß die Funktion nicht mehr auf ganz $\mathbb{R}$ definiert ist.

BEISPIEL:

```
> f:=(x^2+1)/(x^3-1);
```

$$f := \frac{x^2+1}{x^3-1} \tag{2.1.1}$$

```
> simplify(f);
```

$$\frac{x^2+1}{x^3-1} \tag{2.1.2}$$

```
> plot(f,x=0..2);
```




```
> F:=int(f,x);
```

$$F := \frac{2}{3}\ln(x-1) + \frac{1}{6}\ln(x^2+x+1) - \frac{1}{3}\sqrt{3}\,\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right) \qquad \textbf{(2.1.3)}$$

MAPLE kann offenbar besser integrieren, als wir es je von Hand können werden. Trotzdem wollen wir die Probe machen:

```
> diff(F,x);
  simplify(%);
  numer(%)/expand(denom(%));
```

$$\frac{2}{3(x-1)} + \frac{1}{6}\,\frac{2x+1}{x^2+x+1} - \frac{2}{3\left(1+\frac{1}{3}(2x+1)^2\right)}$$

$$\frac{x^2+1}{(x-1)(x^2+x+1)}$$

$$\frac{x^2+1}{x^3-1} \qquad \textbf{(2.1.4)}$$

MATH: Es stellt sich die Frage, ob eine Zerlegung von *f* als Summe, wie wir sie durch Integration und anschließende Differentiation erhalten haben, auch rein algebraisch erhalten werden kann. Das Zauberwort heißt **Partialbruchzerlegung**.

```
> convert(f,parfrac,x);
```

$$\frac{2}{3(x-1)} + \frac{1}{3}\,\frac{x-1}{x^2+x+1} \qquad \textbf{(2.1.5)}$$

Wir haben nur halbes Glück: Nur einer der beiden Terme läßt sich direkt integrieren. Aber auch der andere Term ist recht klar: Damit die Substitutionsregel angewandt werden kann, sorgen wir erst einmal dafür, daß im Zähler so die Ableitung des Nenners steht, dass der Rest ein konstantes

Polynom im Zähler hat. Wir erhalten also einen Summanden der Form $a \cdot \frac{p'}{p}$ und einen Summanden der Form $\frac{b}{p}$ mit $a, b \in \mathbb{R}$ und $p = x^2 + x + 1$.
Von diesen beiden Summanden kann man den ersten sehr leicht integrieren und für Terme in der Form des zweiten Summanden nutzt man man geeigneter Substitution die Identität $\arctan'(x) = \frac{1}{1+x^2}$ zur Integration.

ÜBUNG [04]:

Bestimme die (mit Hilfe von Maple, aber ohne **int**) die Stammfunktion von f auf folgende Art:
1) Führe die oben beschriebene Umformung und Analyse durch, um die gesuchte Zerlegung des Bruches zu erhalten, d.h. spalte den Term $\frac{1}{3}\,\frac{x-1}{x^2+x+1}$ weiter in zwei Summanden auf, wobei in einem der beiden Summanden der Zähler gerade die Ableitung des Nenners ist.
2) Integriere alle drei Summanden, die man aus $\frac{x^2+1}{x^3-1}$ erhalten hat.

Wir wiederholen kurz die Partialbruchzerlegung:

MATH: Es ist besser, die Partialbruchzerlegung für Quotienten von Polynomen statt für die induzierten Funktionen zu erklären:

1.) Schreibe $\frac{p}{q}$ für Polynome *p, q* als Summe eines Polynoms und eines echten Bruches.

(Dieser Schritt entfällt, wenn der Grad von p kleiner als der Grad von q ist.)

BEISPIEL:

```
> f:=(x+1)^3*(x^2+2)/(x^3+x+1);
```

$$f := \frac{(x+1)^3\,(x^2+2)}{x^3+x+1} \tag{2.1.6}$$

```
> numer(f);denom(f);
```

$$(x+1)^3\,(x^2+2)$$
$$x^3+x+1 \tag{2.1.7}$$

```
> quo(numer(f),denom(f),x);
```

$$x^2+3\,x+4 \tag{2.1.8}$$

ist der Polynomanteil und

```
> rem(numer(f),denom(f),x)/denom(f);
```

$$\frac{3x^2 - x - 2}{x^3 + x + 1} \tag{2.1.9}$$

ist der echte Bruchanteil.

MATH: 2.) Benutze den Euklidischen Algorithmus, um den verbleibenden echten Bruch zu kürzen.

3.) Faktorisiere den Nenner in irreduzible Faktoren.
Diese sind bekanntlich (über den reellen Zahlen) vom Grad eins oder zwei.

Hier ist eine gewisse Schwierigkeit: MAPLE kann diese Faktorisierung zunächst nur über $\mathbb{Q}$ durchführen.

```
> factor(x^3+x+1);
```

$$x^3 + x + 1 \tag{2.1.10}$$

Nach dem Zwischenwertsatz sollten wir wenigstens eine reelle Nullstelle haben:

```
> plot(x^3+x+1,x=-1..1);
```



Wir werden ihr den Namen α geben, und MAPLE mit diesem Namen weiterrechnen lassen:

```
> alias(alpha=RootOf(x^3+x+1));
```

$$\alpha \tag{2.1.11}$$

```
> factor(x^3+x+1,alpha);
```

$$-(-x+\alpha)\left(\alpha^2+\alpha x+x^2+1\right) \tag{2.1.12}$$

Da $x^3 + x + 1$ monoton steigend ist, hat es keine weiteren reellen Nullstellen.

MATH: 4.) Ist der Nenner

$$P := \prod_{i=1}^{n} p_i^{a(i)}$$

mit verschiedenen irreduziblen Polynomen p_i, so zerlege den Bruch additiv in die Form

$$\sum_{i=1}^{n} \frac{z_i}{p_i^{a(i)}}$$

wieder mit Hilfe des Euklidischen Algorithmus, welcher auf die

$$Q_i := \frac{P}{p_i^{a(i)}}$$

anzuwenden ist, um Polynome R_i zu finden mit

$$\sum_{i=1}^{n} R_i Q_i = 1.$$

ÜBUNG [05]:

Führe Schritt 4.) für den vorliegenden Fall

$$\frac{-2 + 3\cdot x^2 - x}{(x-\alpha)\cdot\left(x^2 + \alpha x + 1 + \alpha^2\right)}$$

durch. (Hinweis: Benutze gcdex)

```
> (-2+3*x^2-x)/((x-alpha)*(x^2+alpha*x+1+alpha^2));
```

$$\frac{3x^2 - x - 2}{(x-\alpha)\left(\alpha^2 + \alpha x + x^2 + 1\right)} \quad \textbf{(2.1.13)}$$

MATH: 5.) Nach erneuter Ausführung von Schritt 1) kann man ohne Einschränkung davon ausgehen, dass jeder Summand

$$\frac{z}{p^a} := \frac{z_i}{p_i^{a(i)}}$$

$Grad(z) < Grad(p^a)$ erfüllt. Schreibe diesen Summanden als

$$\frac{z}{p^a} = \frac{A_1}{p} + \frac{A_2}{p^2} + \ldots + \frac{A_a}{p^a}$$

im Falle $p = x - w \in \mathbb{R}[x]$ vom Grad 1 mit $A_j \in \mathbb{R}$

und

$$\frac{z}{p^a} = \frac{B_1 + x\cdot C_1}{p} + \frac{B_2 + x\cdot C_2}{p^2} + \ldots + \frac{B_a + x\cdot C_a}{p^a}$$

im Falle $p = x^2 + v\cdot x + u \in \mathbb{R}[x]$ vom Grad 2 mit $B_j, C_j \in \mathbb{R}$.

Unsere Funktion f war so einfach gewählt, dass dieser Schritt nicht notwendig wird.

Wir führen dies an an einem anderen Beispiel durch, so dass klar wird, wie dies allgemein geht:

```
> s:=add(x^i,i=0..5)/(x-2)^6;
```

$$s := \frac{x^5 + x^4 + x^3 + x^2 + x + 1}{(x-2)^6} \quad (2.1.14)$$

```
> expand(subs(x=y+2,numer(s)));
```

$$y^5 + 11\,y^4 + 49\,y^3 + 111\,y^2 + 129\,y + 63 \quad (2.1.15)$$

```
> subs(y=``(x-2),%);
```

$$(x-2)^5 + 11\,(x-2)^4 + 49\,(x-2)^3 + 111\,(x-2)^2 + 129\,(x-2) + 63 \quad (2.1.16)$$

Damit ist klar, wie es sich im linearen Fall verhält. Man hätte auch schneller den Taylorbefehl benutzen können:

```
> taylor(numer(s),x=2);
```

$$63 + 129\,(x-2) + 111\,(x-2)^2 + 49\,(x-2)^3 + 11\,(x-2)^4 + (x-2)^5 \quad (2.1.17)$$

Beachte: Die obigen Koeffizienten 63, 129, 111 etc. erhält man als den Reste: 63 als Rest von *numer(s)* modulo *(x-2)*. Man subtrahiert 63, dividiert durch *(x-2)* und erhält 129 als den Rest *mod (x-2)* etc.

freiwillige ÜBUNG:

Führe die oben beschriebene Zerlegung von Schritt 5.) durch für

```
> t:=add(x^i,i=0..5)/(x^2+x+2)^4;
```

$$t := \frac{x^5 + x^4 + x^3 + x^2 + x + 1}{(x^2 + x + 2)^4} \quad (2.1.18)$$

Wir haben in diesem Abschnitt bereits (fast) alles gemacht um $f = \frac{(x+1)^3 \cdot (x^2+2)}{x^3 + x + 1}$ zu integrieren.

DENKANSTOSS: Führe die Partialbruchzerlegung auf die Bestimmung der Faktorisierung des Nenners und das Lösen eines LGS zurück. Warum wurde sie im Worksheet anders eingeführt? (*Hinweis*: Koeffizientenvergleich)

ÜBUNG [06]:

1) Fasse zusammen, welche Zerlegungen von $f = \frac{(x+1)^3 \cdot (x^2+2)}{x^3 + x + 1}$ in Summanden bereits gemacht wurden, um die Integration vorzubereiten.
2) Zerlege f weiter in eine Summe von Termen, bis du alle Summanden (von Hand) integrieren könntest (zum Beispiel mit den in Aufgabe 1 verwendeten Integrationsverfahren).
3) Integriere alle Summanden. (Es kann zwar der **int**-Befehl benutzt werden, aber wir erwarten, dass du genau beschreiben kannst, welche Integrationsregeln aus welchem Grund angewendet werden.)

```
> f ;
```

$$\frac{(x+1)^3 (x^2+2)}{x^3+x+1} \tag{2.1.19}$$

MATH: Wir wollen uns noch davon überzeugen, dass man (oder MAPLE) auch s und t leicht integrieren kann:

```
> int(s,x);
```

$$-\frac{49}{2(x-2)^2}-\frac{37}{(x-2)^3}-\frac{63}{5(x-2)^5}-\frac{11}{x-2}+\ln(x-2)-\frac{129}{4(x-2)^4} \tag{2.1.20}$$

```
> int(t,x);
```

$$\frac{1}{7}\frac{-3x-5}{x^2+x+2}-\frac{2}{343}\sqrt{7}\arctan\left(\frac{1}{7}(2x+1)\sqrt{7}\right)+\frac{1}{21}\frac{-7x-7}{(x^2+x+2)^3} \tag{2.1.21}$$
$$-\frac{5}{42}\frac{2x+1}{(x^2+x+2)^2}+\frac{10}{49}\frac{2x+1}{x^2+x+2}+\frac{1}{14}\frac{10x+12}{(x^2+x+2)^2}$$

Bei t kommt wieder der anfangs behandelte Trick mit der quadratischen Ergänzung zum Tragen, allerdings in iterierter Form.

DENKANSTOSS: Man könnte komplex vorgehen und hätte dann eine einfachere Partialbruchzerlegung. Man würde sich aber komplexe Logarithmen einhandeln.

MATH: Zur Integration von rationalen Funktionen fehlt noch die Stammfunktion von $\frac{1}{(x^2+1)^n}$, da wir alle anderen Summanden der Partialbruchzerlegung mit den obigen Methoden im Griff haben. Dafür gibt es eine rekursive Formel.

Uneigentliche Integrale

Aufgaben: 3

```
> restart;
```

Uneigentliche Integrale

Das Integral ist bereits ein Grenzwert. Kann man die obere Grenze gegen unendlich oder die untere gegen -unendlich laufen lassen oder eines von beiden gegen eine Unendlichkeitsstelle (Polstelle) der Funktion, so kann man wieder Grenzwertbetrachtungen anstellen. Diese sind für sich interessant, lassen sich aber auch auf Konvergenzbetrachtungen von Reihen anwenden, weil der Fundamentalsatz oft die explizite Bestimmung der Integralfunktionen ermöglicht.

MATH: Es gibt diverse Situationen, wo man über uneigentliche Riemannintegrierbarkeit spricht. Wir wollen erstmal nur eine hervorheben:

$$f\colon [a,\infty) \to \mathbb{R}$$

heißt uneigentlich Riemannintegrierbar, falls die Einschränkung von *f* auf jedes Intervall [*a*,*b*] mit $b > a$ Riemannintegrierbar ist und

$$\lim_{b\to\infty} \int_a^b f(x)\,dx$$

existiert (als reelle Zahl).

MAPLE kann uneigentliche Riemannintegrierbarkeit oftmals leicht feststellen, z. B. weil es die Stammfunktion formal ausrechnen kann.

Wir haben auch schon ein paar Integrationstechniken kennengelernt und wollen diese hier anwenden.

ÜBUNG [07]:

Bei dieser Aufgabe soll ausdrücklich *nicht* der **Int**-Befehl von Maple benutzt werden. Der **limit**-Befehl darf aber verwendet werden.

1) Sei $t \in \left(0, \frac{\pi}{2}\right)$. Zeige: $\cos(t) = \frac{1}{\sqrt{\tan(t)^2 + 1}}$.

2) Folgere: $\cos(\arctan(x)) = \frac{1}{\sqrt{x^2+1}}$ für $x > 0$.

Welche der folgenden Funktionen $f\colon [7\cdot\pi, \infty) \to \mathbb{R}$ sind uneigentlich Riemannintegrierbar (für $x \to \infty$)?

3) $\arctan(x)$ (Hinweis: Integration der Umkehrfunktion)

4) $f(x) = \frac{\pi}{2} - \arctan(x)$

5) $f(x) = \frac{1}{x^2+1}$

6) $f(x) = \frac{1}{x^2-1}$ (Hinweis: Partialbruchzerlegung)

Integralvergleichskriterium für Reihen

MATH: (Integralvergleichskriterium für Reihen): Sei

$$f\colon [1,\infty) \to \mathbb{R}$$

monoton fallend. Dann sind

$$\sum_{k=1}^{\infty} f(k)$$

und

$$\int_1^{\infty} f(x)\,dx$$

entweder beide divergent oder beide konvergent.

Die Anwendung ist meistens so, dass man die Konvergenz des Integrals entscheiden kann und dann auf die Reihe schließt. Der Grenzwert der Reihe ist im Falle der Konvergenz dann aber immer noch nicht bestimmt.

ÜBUNG [08]:

1) Ist $\frac{\ln(n)}{n^2}$ monoton fallend?

2) Benutze das obige Vergleichskriterium, um die Konvergenz von

$$\sum_{n=1}^{\infty} \frac{\ln(n)}{n^2}$$

zu entscheiden.

Eulersche Integrale

Hier geht es um die Eulersche Betafunktion und ihren Zusammenhang mit der Gammafunktion, welche die Fakultätsfunktion interpoliert. Die Betafunktion ist verwandt mit dem Kehrwert der Binomialkoeffizienten:

```
> Beta(p,q)=Int(x^(p-1)*(1-x)^(q-1),x=0..1);
```

$$\mathrm{B}(p, q) = \int_0^1 x^{p-1} (1-x)^{q-1}\,\mathrm{d}x \qquad \textbf{(3.3.1)}$$

```
> Beta(1,q);
```

$$\frac{1}{q} \qquad \textbf{(3.3.2)}$$

(Das griechische "grosse β" sieht aus wie ein B.)

ÜBUNG [09]:

1) Für welche (reellen) Werte von p und q ist der obige Ausdruck $\mathrm{B}(p, q)$ ein uneigentliches Integral?
2) Wann ist es divergent?
Hinweis: Unterteile das Intervall $[0, 1]$ und schätze jeweils einen Faktor des Integranden ab.

MATH: Die Gammafunktion

$$\Gamma : x \mapsto \int_0^\infty t^{x-1} \mathrm{e}^{-t}\, \mathrm{d}t$$

```
> int(t^(x-1)*exp(-t),t=0..infinity);
```

$$\Gamma(x) \tag{3.3.3}$$

interpoliert die Funktion

$$\mathbb{N} \to \mathbb{N} : n \to (n-1)!\ ,$$

wie man durch iterierte partielle Integration sieht.

```
> GAMMA(1),0!;
  GAMMA(2),1!;
  GAMMA(3),2!;
  GAMMA(4),3!;
  GAMMA(5),4!;
```

$$1,\ 1$$
$$1,\ 1$$
$$2,\ 2$$
$$6,\ 6$$
$$24,\ 24 \tag{3.3.4}$$

Man kann B mit hilfe von Γ darstellen

```
> int(x^(p-1)*(1-x)^(q-1),x=0..1);
```

$$\frac{\Gamma(q)\,\Gamma(p)}{\Gamma(p+q)} \tag{3.3.5}$$

und erkennt die Ähnlichkeit zum Kehrwert des Binomialkoeffizienten.